Torsion Wave Animation

The left end, $\theta_0(t)$ is fixed and equal to zero. I am not even going to bother writing its equation. Also, I will introduce $\omega_i(t) = d\theta_i/dt$ so that we don't have second-order differential equations. Instead we will have twice as many equations, but with only one time derivative.

In[235]:=

```
n = 71; (* number of rods, not counting the fixed one *)
v = 1; (* velocity *)
tmax = 250; (* max time that simulation will run *)
pulseleft = 44;
pulsewidth = 24;
pulseright = pulseleft + pulsewidth;
```

In[241]:=

```
positionEquations = {Table[θ[i]'[t] == ω[i][t], {i, 1, n}]};
```

In[242]:=

```
initialPositions = {Table[θ[i][0] == If[
     i < pulseleft || i > pulseright, 0, Sin[2 (i - pulseleft) Pi / 24]], {i, 1, n}]};
```

In[243]:=

```
momentumEquations = {Table[ω[i]'[t] ==
    v^2 (If[i == n, 0, θ[i + 1][t] - θ[i][t]] - (θ[i][t] - If[i == 1, 0, θ[i - 1][t]])), {i,
    1, n}]};
```

In[244]:=

```
initialMomenta = {Table[ω[i][0] == If[i < pulseleft || i ≥ pulseright,
     0, 2 π / 24 * Cos[2 (i - pulseleft) Pi / pulsewidth]], {i, 1, n}]};
```

In[245]:=

```
interpolatingFunctions = NDSolve[Flatten[{positionEquations,
    momentumEquations, initialPositions, initialMomenta}, 1],
   Flatten[{Table[θ[i], {i, 1, n}], Table[ω[i], {i, 1, n}]}, 1], {t, tmax}];
```

In[246]:=

```
plots = Table[ListPlot[
    Table[{i, If[i == 0, 0, θ[i][t]]}, {i, 0, n}] /. interpolatingFunctions,
    PlotRange → {{0, n}, {-2.2, 2.2}}, ImageSize → Large], {t, 0, tmax, 0.1}];
```

In[247]:=

```
Animate[plots[[a]], {a, 1, 10 * tmax, 1},
 AnimationRepetitions → 1, AnimationRate → 60]
```

Out[247]=



A torsion wave demonstration with a real apparatus with 72 rods (produced by Pasco Scientific): https://youtu.be/MrZcMTLK6W4?t=23